# Maintainer: Felix Yan <felixonmars@archlinux.org>

_hkgname=html-conduit
pkgname=haskell-html-conduit
pkgver=1.3.2.2
pkgrel=158
pkgdesc="Parse HTML documents using xml-conduit datatypes"
url="https://github.com/snoyberg/xml"
license=('MIT')
arch=('loong64' 'x86_64')
depends=('ghc-libs' 'haskell-resourcet' 'haskell-conduit' 'haskell-xml-conduit'
         'haskell-attoparsec' 'haskell-conduit-extra' 'haskell-xml-types')
makedepends=('ghc' 'haskell-hspec' 'haskell-hunit')
source=(https://hackage.haskell.org/packages/archive/${_hkgname}/${pkgver}/${_hkgname}-${pkgver}.tar.gz)
sha512sums=('474bfeec36d0f2fe8eb37ed88372d456800e8d28e499e5ffe222170db794eba292938800e145613334d1e11d2e6b78195b6dd7a0d3cd92e2fa3f348ad8d18489')

build() {
    cd $_hkgname-$pkgver
    
    runhaskell Setup configure -O --enable-shared --enable-executable-dynamic --disable-library-vanilla \
        --prefix=/usr --docdir=/usr/share/doc/$pkgname --datasubdir=$pkgname --enable-tests \
        --dynlibdir=/usr/lib --libsubdir=\$compiler/site-local/\$pkgid --ghc-option=-fllvm
    runhaskell Setup build $MAKEFLAGS
    runhaskell Setup register --gen-script
    runhaskell Setup unregister --gen-script
    sed -i -r -e "s|ghc-pkg.*update[^ ]* |&'--force' |" register.sh
    sed -i -r -e "s|ghc-pkg.*unregister[^ ]* |&'--force' |" unregister.sh
}

check() {
    cd $_hkgname-$pkgver
    runhaskell Setup test --show-details=direct
}

package() {
    cd $_hkgname-$pkgver
    
    install -D -m744 register.sh "$pkgdir"/usr/share/haskell/register/$pkgname.sh
    install -D -m744 unregister.sh "$pkgdir"/usr/share/haskell/unregister/$pkgname.sh
    runhaskell Setup copy --destdir="$pkgdir"
    install -D -m644 LICENSE "$pkgdir"/usr/share/licenses/$pkgname/LICENSE
    rm -f "$pkgdir"/usr/share/doc/$pkgname/LICENSE
}
